

# SIM808 GPS/GPRS/GSM Shield SKU: TEL0097

## Contents

- 1 Introduction
- 2 Specification
- 3 Board Overview
- 4 Arduino Examples
  - 4.1 Preparation
  - 4.2 Examples
    - 4.2.1 Make Phone Calls
    - 4.2.2 Send SMS Messages
    - 4.2.3 Auto Answering Phone Calls and Reading SMS Messages
    - 4.2.4 Read SMS Messages
    - 4.2.5 Connect TCP and Send GET Requests
    - 4.2.6 Get GPS Data
- 5 USB Debugging (AT command)
  - 5.1 Preparation
  - 5.2 Make a Phone Call
  - 5.3 Answer the phone and Hang up
  - 5.4 Send SMS
  - 5.5 Read SMS
  - 5.6 TCP Communication
  - 5.7 GPS Orientation
- 6 SIM808 Low Power Consumption Mode
- 7 Minimum Functionality Mode
- 8 Instructions
- 9 More



(/wiki/index.php/File:TEL0097\_SIM808.jpg)

SIM808 GPS/GPRS/GSM Shield

## Introduction

SIM808 GPS/GPRS/GSM is an integrated quad-band GSM/GPRS and GPS navigation technology Arduino expansion shields. A credit card size only, according to the standard Arduino pin packaging, compatible with Arduino UNO, Leonardo, Mega and other mainboard.


Compared to the previous generation SIM908, SIM808 made some improvement on the performance and stability. In addition to the normal SMS and phone functions, the shield also supports MMS, DTMF, FTP and other functions. You can achieve the data acquisition, wireless data transceiver, IoT application and GPS orientating. The shield integrates onboard microphone and headphone jack, saving your cost and making your project easily. It can also directly connect to the GSM and GPS antenna by an external antenna connector.

SIM808 GPS/GPRS/GSM Shield V1.0 uses the latest version of Simcom SIM808 module, compared with the early SIM808 module available in the market, the new module has better stability. But GPS part of AT commands are not compatible with the old version of the SIM808 module, please refer to the bottom of the "more" in the AT commands.

## Specification

- Operating voltage: 5V
- Input Power: 7-23V
- Quad-band 850/900/1800/1900MHz
- GPRS multi-slot class 12/10
- GPRS mobile station class B
- Comply with GSM phase 2/2 +
  - Class 4 (2 W @ 850 / 900MHz)
  - Class 1 (1 W @ 1800 / 1900MHz)
- Support low power consumption mode: 100mA @ 7V-GSM mode
- Support AT command control (3GPP TS 27.007, 27.005 and SIMCOM enhanced AT Commands)
- Support GPS satellite navigation technology
- Support LED status indicator: Power supply status, network status and operating modes
- Working environment: -40 °C ~ 85 °C
- Size: 69 \* 54mm/2.71 \* 2.12 inches

## Board Overview



#### Module interface and precautions:

1. Occupied pins: D0, D1, D12, the digital pin "D12" is connected to SIM808 module power GPIO. It can be used as SIM808 module on / off control.
2. SIM808 onboard MIC and 3.5mm SIM808 microphone are using the same MIC channel, when you plugin your microphone, the on-board MIC will be automatically disconnected.
3. Boot switch button --SIM808 module switch, short-press 1s to start SIM808 , long-press 3s turn it off.
4. LED "ON" - SIM808 power indicator, only when you connect an external power, the module can work properly.
5. Net - GSM signal indicator
  1. Fast Flash: Search Network
  2. Slow flash (3s once): network registration complete
6. Function switch
  1. None-- Vacant SIM808 serial pin to download the sketch, please dial to here.
  2. USB\_DBG-- When the expansion board plugged on the Arduino, dial to here to make SIM808 communicate with PC to make a debug (AT debugging).
  3. Arduino-- When the expansion board plugged on the Arduino, dial to here to make SIM808 communicate with Arduino.

## Arduino Examples

In this tutorial we will cover how to use Arduino to drive a SIM808 GPS/GPRS/GSM Shield. DFRobot\_SIM808 is a open-source Arduino library with intuitive functions. With an array of functions packaged into a single library, you can focus on your project and not waste time studying complex AT commands.

- This tutorial includes:
  - Send and receive GPRS data
  - Receive GPS data
  - Send and receive SMS messages
  - Make and receive phone calls

## Preparation

- Hardware:
  - 1 x Arduino UNO
  - 1 x SIM808 Expansion Shield
  - 1 x External Power Supply
  - 1 x SIM Card
- Software:
  - Arduino IDE Click to Download Arduino IDE from Arduino® (<https://www.arduino.cc/en/Main/Software>)
  - DFRobot\_SIM808 library Click to Download DFRobot\_SIM808 library ([https://github.com/DFRobot/DFRobot\\_SIM808](https://github.com/DFRobot/DFRobot_SIM808))
- Steps:
  1. Insert a SIM card in to the SIM slot on the SIM808 expansion shield
  2. Stack the expansion shield on to an Arduino UNO
  3. Connect an external power source to the Arduino
  4. Set the function switch to 'None'

5. Upload the sample code
6. Set the function switch to `Arduino` and make sure SIM808 could communicate with Arduino board
7. Press the Boot power button
8. Wait for the SIM card to register the network, the Net indicator LED will slowly flash every 3 seconds

## Examples

### Make Phone Calls

This example tests DFRobot\_SIM808 SIM808 GPS/GPRS/GSM Shield's callUp function.

1. Open the SIM808\_CallUp example or copy the following code to your IDE
2. Input your SIM's phone number in the line: ``#define PHONE\_NUMBER "187\*\*\*\*\*39"``
3. Upload the code and set the function switch to `Arduino`

```

1  #include <DFRobot_sim808.h>
2
3  //Mobile phone number, need to change
4  #define PHONE_NUMBER "187*****39"
5
6  DFRobot_SIM808 sim808(&Serial);
7
8  void setup() {
9      //mySerial.begin(9600);
10     Serial.begin(9600);
11     //*****Initialize sim808 module*****
12     while(!sim808.init()) {
13         delay(1000);
14         Serial.print("Sim808 init error\r\n");
15     }
16     Serial.println("Sim808 init success");
17     Serial.println("Start to call ...");
18
19     //*****Call specified number*****
20     sim808.callUp(PHONE_NUMBER);
21 }
22 void loop() {
23     //nothing to do
24 }
```

### Send SMS Messages

This example tests the DFRobot\_SIM808 GPS/GPRS/GSM shield's send SMS functions

1. Open the SIM808\_SendSMS example or copy the code to your project
2. Input your SIM's phone number in the line: ``#define PHONE\_NUMBER "187\*\*\*\*\*39"``
3. Set the function switch to `Arduino`

```

1  #include <DFRobot_sim808.h>
2
3  //Mobile phone number, need to change
4  #define PHONE_NUMBER "187*****39"
5
6  //The content of messages sent
7  #define MESSAGE "hello,world"
8
9  DFRobot_SIM808 sim808(&Serial);
10
11 void setup() {
12     //mySerial.begin(9600);
13     Serial.begin(9600);
14
15     //***** Initialize sim808 module *****
16     while(!sim808.init()) {
17         delay(1000);
18         Serial.print("Sim808 init error\r\n");
19     }
20     Serial.println("Sim808 init success");
21     Serial.println("Start to send message ...");
22
23     //***** define phone number and text *****
24     sim808.sendSMS(PHONE_NUMBER,MESSAGE);
25 }
26
27 void loop() {
28     //nothing to do
29 }
```

### Auto Answering Phone Calls and Reading SMS Messages

This example tests the DFRobot\_SIM808 GPS/GPRS/GSM Shield's call answering and SMS reading function.

1. Open the 'SIM808\_LoopHandle' example or copy the code to your project
2. Download and set the function switch to 'Arduino'
3. The shield will auto-answer phone calls or send SMS Messages to the serial terminal

```

1 #include <DFRobot_sim808.h>
2
3 #define MESSAGE_LENGTH 20
4 char gprsBuffer[64];
5 char *s = NULL;
6
7 DFRobot_SIM808 sim808(&Serial);
8
9 void setup() {
10     //mySerial.begin(9600);
11     Serial.begin(9600);
12
13     //***** Initialize sim808 module *****
14     while(!sim808.init()) {
15         Serial.print("Sim808 init error\r\n");
16         delay(1000);
17     }
18     delay(3000);
19     Serial.println("Init Success, please call or send SMS message to me!");
20 }
21
22 void loop() {
23     //***** Wait serial data *****
24     if(sim808.readable()){
25         sim808_read_buffer(gprsBuffer,32,DEFAULT_TIMEOUT);
26         //Serial.print(gprsBuffer);
27
28         //***** Detect the current state of the telephone or SMS *****
29         if(NULL != strstr(gprsBuffer,"RING")) {
30             sim808.answer();
31         }else if(NULL != (s = strstr(gprsBuffer,"+CMTI: \\"SM\\"))) { //SMS: $$+CMTI: "SM",24$$
32             char message[MESSAGE_LENGTH];
33             int messageIndex = atoi(s+12);
34             sim808.readSMS(messageIndex, message,MESSAGE_LENGTH);
35             Serial.print("Recv Message: ");
36             Serial.println(message);
37         }
38         sim808_clean_buffer(gprsBuffer,32);
39     }
40 }

```

## Read SMS Messages

This example tests the DFRobot\_SIM808 GPS/GPRS/GSM Shield's ability to read SMS messages.

1. Open the 'SIM808\_SMSread' example or copy the code to your project
2. Download and set the function switch to 'Arduino'
3. The shield will receive SMS Messages and send them to the serial terminal

```

1 #include <DFRobot_sim808.h>
2
3 #define MESSAGE_LENGTH 160
4 char message[MESSAGE_LENGTH];
5 int messageIndex = 0;
6
7 char phone[16];
8 char datetime[24];
9
10 DFRobot_SIM808 sim808(&Serial);
11
12 void setup() {
13   //mySerial.begin(9600);
14   Serial.begin(9600);
15
16   //***** Initialize sim808 module *****
17   while(!sim808.init()) {
18     Serial.print("Sim808 init error\r\n");
19     delay(1000);
20   }
21   delay(3000);
22   Serial.println("Init Success, please send SMS message to me!");
23 }
24
25 void loop() {
26   //***** Detecting unread SMS *****
27   messageIndex = sim808.isSMSUnread();
28   Serial.print("messageIndex: ");
29   Serial.println(messageIndex);
30
31   //***** At Least, there is one UNREAD SMS *****
32   if (messageIndex > 0) {
33     sim808.readSMS(messageIndex, message, MESSAGE_LENGTH, phone, datetime);
34
35     //*****In order not to full SIM Memory, is better to delete it*****
36     sim808.deleteSMS(messageIndex);
37     Serial.print("From number: ");
38     Serial.println(phone);
39     Serial.print("Datetime: ");
40     Serial.println(datetime);
41     Serial.print("Recieved Message: ");
42     Serial.println(message);
43   }
44 }

```

## Connect TCP and Send GET Requests

This example tests the DFRobot\_SIM808 GPS/GPRS/GSM Shield's ability to connect TCP and send GET requests

1. Open the `SIM808\_TCPConnection` example or copy the code to your project
2. Download and set the function switch to `Arduino`
3. Open the serial terminal
4. Wait for until `Connect mbed.org success` is printed in the terminal
5. The serial terminal will print `Hello world!`

```

1  #include <DFRobot_sim808.h>
2
3  //make sure that the baud rate of SIM900 is 9600!
4  //you can use the AT Command(AT+IPR=9600) to set it through SerialDebug
5
6  DFRobot_SIM808 sim808(&Serial);
7
8  char http_cmd[] = "GET /media/uploads/mbed_official/hello.txt HTTP/1.0\r\n\r\n";
9  char buffer[512];
10
11 void setup(){
12     //mySerial.begin(9600);
13     Serial.begin(9600);
14
15     //***** Initialize sim808 module *****
16     while(!sim808.init()) {
17         delay(1000);
18         Serial.print("Sim808 init error\r\n");
19     }
20     delay(3000);
21
22     //***** Attempt DHCP *****
23     while(!sim808.join(F("cmnet"))){
24         Serial.println("Sim808 join network error");
25         delay(2000);
26     }
27
28     //***** Successful DHCP *****
29     Serial.print("IP Address is ");
30     Serial.println(sim808.getIPAddress());
31
32     //***** Establish a TCP connection *****
33     if(!sim808.connect(TCP,"mbed.org", 80)) {
34         Serial.println("Connect error");
35     }else{
36         Serial.println("Connect mbed.org success");
37     }
38
39     //***** Send a GET request *****
40     Serial.println("waiting to fetch...");
41     sim808.send(http_cmd, sizeof(http_cmd)-1);
42     while (true) {
43         int ret = sim808.recv(buffer, sizeof(buffer)-1);
44         if (ret <= 0){
45             Serial.println("fetch over...");
46             break;
47         }
48         buffer[ret] = '\0';
49         Serial.print("Recv: ");
50         Serial.print(ret);
51         Serial.print(" bytes: ");
52         Serial.println(buffer);
53         break;
54     }
55
56     //***** Close TCP or UDP connections *****
57     sim808.close();
58
59     //*** Disconnect wireless connection, Close Moving Scene *****
60     sim808.disconnect();
61 }
62
63 void loop(){
64
65 }

```

## Get GPS Data

This example tests SIM808 GPS/GPRS/GSM Shield's ability to read GPS data.

1. Open the 'SIM808\_GetGPS' example or copy the code to your project
2. Download and set the function switch to 'Arduino'
3. Open the serial terminal
4. Place the shield outside, wait for a few minutes and it will send GPS data to serial terminal

```

1  #include <DFRobot_sim808.h>
2
3  DFRobot_SIM808 sim808(&Serial);
4
5  void setup() {
6      //mySerial.begin(9600);
7      Serial.begin(9600);
8
9      //***** Initialize sim808 module *****
10     while(!sim808.init()) {
11         delay(1000);
12         Serial.print("Sim808 init error\r\n");
13     }
14
15     //***** Turn on the GPS power*****
16     if( sim808.attachGPS())
17         Serial.println("Open the GPS power success");
18     else
19         Serial.println("Open the GPS power failure");
20
21 }
22
23 void loop() {
24     //***** Get GPS data *****
25     if (sim808.getGPS()) {
26         Serial.print(sim808.GPSdata.year);
27         Serial.print("/");
28         Serial.print(sim808.GPSdata.month);
29         Serial.print("/");
30         Serial.print(sim808.GPSdata.day);
31         Serial.print(" ");
32         Serial.print(sim808.GPSdata.hour);
33         Serial.print(":");
34         Serial.print(sim808.GPSdata.minute);
35         Serial.print(":");
36         Serial.print(sim808.GPSdata.second);
37         Serial.print(":");
38         Serial.println(sim808.GPSdata.centisecond);
39         Serial.print("latitude :");
40         Serial.println(sim808.GPSdata.lat);
41         Serial.print("longitude :");
42         Serial.println(sim808.GPSdata.lon);
43         Serial.print("speed_kph :");
44         Serial.println(sim808.GPSdata.speed_kph);
45         Serial.print("heading :");
46         Serial.println(sim808.GPSdata.heading);
47         Serial.println();
48
49         //***** Turn off the GPS power *****
50         sim808.detachGPS();
51     }
52
53 }

```

## USB Debugging (AT command)

In this section, we will send AT commands through the serial port to debug SIM808 expansion shield. Please refer to SIM808 AT command set for more function.

### Preparation


- Hardware:
  - Arduino UNO x1
  - SIM808 expansion board x1
  - External power supply x1
- Software :
  - Arduino IDE Click to Download Arduino IDE from Arduino® (<https://www.arduino.cc/en/Main/Software>)
  - Serial debugging assistant (In this section, we are using DF Serial Debugger by Lisper ([https://github.com/leffhub/DFRobotSIM808\\_Leonardo\\_mainboard/raw/master/Software/DF%20Serial%20Debugger.exe](https://github.com/leffhub/DFRobotSIM808_Leonardo_mainboard/raw/master/Software/DF%20Serial%20Debugger.exe)))

“STEP”

1. Plug your SIM card on the SIM808 expansion shield and plug the expansion shield on Arduino UNO, meanwhile don't forget to connect an external power source.
2. Dial the function switch to **None**, download a sample code Blink to make sure the serial port is not occupied.

3. Press the Boot power button and wait for the SIM card registering a network properly, the **Net** indicator led flashes slowly (3s once).
4. Slide the function switch to **USB\_DBG**, then we can communicate directly with SIM808 chip through the serial port assistant.

## Make a Phone Call




(/wiki/index.php/File:TEL0097\_Phone\_call.png)

Make a Phone Call

Send AT in the serial port assistant, if it returns **OK**, it means that serial communication has been established. Follow the Figure, enter AT commands, you should get the same content.


## Answer the phone and Hang up




(/wiki/index.php/File:TEL0097\_Answer\_Hang\_up.png)

Answer the phone and Hang up

## Send SMS








(/wiki/index.php/File:TEL0097\_Send\_Message.png)

Send Message


## Read SMS




(/wiki/index.php/File:TEL0097\_Read\_Message.png)

Read Message

## TCP Communication





If you are going to use SIM808 GPS/GPRS/GSM Shield with a Leonardo board, it will be more like a SIM808 GPS/GPRS/GSM mainboard ([http://www.dfrobot.com/index.php?route=product/product&product\\_id=1448&search=sim808&description=true#.V0QaxvmqqAw](http://www.dfrobot.com/index.php?route=product/product&product_id=1448&search=sim808&description=true#.V0QaxvmqqAw)). You can check its wiki for the user manual.

## More

SIM808 GPS/GPRS/GSM Shield schematic

([https://github.com/Arduinolibrary/DFRobot\\_GPS\\_GPRS\\_GSM\\_shield/raw/master/TEL0097%20SIM808%20%20GPSGPRSGSM%20Shield\(V1.0\)%E5%8E%9F%E7%90%99](https://github.com/Arduinolibrary/DFRobot_GPS_GPRS_GSM_shield/raw/master/TEL0097%20SIM808%20%20GPSGPRSGSM%20Shield(V1.0)%E5%8E%9F%E7%90%99)

NEW SIM800 Series GNSS Application Note V1.00

([https://github.com/leffhub/DFRobotSIM808\\_Leonardo\\_mainboard/raw/master/Documents/NEW%20SIM800%20Series\\_GNSS\\_Application%20Note%20V1.00.pdf](https://github.com/leffhub/DFRobotSIM808_Leonardo_mainboard/raw/master/Documents/NEW%20SIM800%20Series_GNSS_Application%20Note%20V1.00.pdf))

SIM800 Series AT Command Manual V1.07

([https://github.com/leffhub/DFRobotSIM808\\_Leonardo\\_mainboard/blob/master/SIM800\\_Series\\_AT\\_Command\\_Manual\\_V1.07.pdf](https://github.com/leffhub/DFRobotSIM808_Leonardo_mainboard/blob/master/SIM800_Series_AT_Command_Manual_V1.07.pdf))

SIM808 Hardware Design\_V1.00

([https://github.com/leffhub/DFRobotSIM808\\_Leonardo\\_mainboard/raw/master/Documents/SIM808\\_Hardware%20Design\\_V1.00.pdf](https://github.com/leffhub/DFRobotSIM808_Leonardo_mainboard/raw/master/Documents/SIM808_Hardware%20Design_V1.00.pdf))

SVG file ([https://github.com/Arduinolibrary/DFRobot\\_GPS\\_GPRS\\_GSM\\_shield/raw/master/TEL0097\\_EN.svg](https://github.com/Arduinolibrary/DFRobot_GPS_GPRS_GSM_shield/raw/master/TEL0097_EN.svg))



(<http://www.dfrobot.com/>) Shopping from **DFRobot Store** (<http://www.dfrobot.com/index.php>) or **DFRobot Distributor**.  
(<http://www.dfrobot.com/index.php?route=information/distributorslogo>)

This page was last modified on 29 September 2016, at 11:37.

Content is available under GNU Free Documentation License 1.3 or later (<https://www.gnu.org/copyleft/fdl.html>) unless otherwise noted.



(<https://www.gnu.org/copyleft/fdl.html>)



(<http://www.mediawiki.org/>)